
  Intro


Live config supporting many different backends.
Kungfuig (pronounced: [ˌkʌŋˈfig]) provides an easy way to plug
live configuration into everything.
It provides backends for env and system and supports custom backends. Configurations are periodically updated, and changes can trigger callback notifications to relevant application components.

  
    
  
  Installation


def deps do
  [
    {:kungfuig, "~> 1.0"}
  ]
end

  
    
  
  Architecture


Kungfuig is designed with a robust architecture that provides flexibility and reliability:
[image: Kungfuig Architecture]
The architecture consists of several key components:
	Supervisor (Kungfuig.Supervisor): Manages the configuration process tree
	Manager (Kungfuig.Manager): Supervises the backend workers
	Backend Workers (Kungfuig.Backend implementations): Poll configuration sources
	Blender (Kungfuig.Blender): Combines configurations from all backends
	Callbacks: Notify application components when configurations change
	Validators: Ensure configuration values match expected schemas

This architecture provides several benefits:
	Fault tolerance through supervisor patterns
	Separation of concerns between configuration sources
	Unified configuration access for your application
	Pluggable backends for custom configuration sources


  
    
  
  Using



  
    
  
  Basic Usage


Kungfuig is the easy way to read the external configuration from sources that are not controlled by the application using it, such as Redis, or Database.
# Start Kungfuig with default options
Kungfuig.start_link()

# Get all configuration
Kungfuig.config()
#⇒ %{env: %{kungfuig: []}, system: %{}}

# Get configuration from a specific backend
Kungfuig.config(:env)
#⇒ %{kungfuig: []}

# Update application environment and see the change
Application.put_env(:kungfuig, :foo, 42)
Kungfuig.config(:env)
#⇒ %{kungfuig: [foo: 42]}

  
    
  
  Custom Backends


Here is an example of a backend implementation for reading configuration from an external MySQL database:
defmodule MyApp.Kungfuig.MySQL do
  @moduledoc false

  use Kungfuig.Backend, interval: 300_000 # 5 minutes

  @impl Kungfuig.Backend
  def get(_meta) do
    with {:ok, host} <- System.fetch_env("MYSQL_HOST"),
         {:ok, db} <- System.fetch_env("MYSQL_DB"),
         {:ok, user} <- System.fetch_env("MYSQL_USER"),
         {:ok, pass} <- System.fetch_env("MYSQL_PASS"),
         {:ok, pid} when is_pid(pid) <-
           MyXQL.start_link(hostname: host, database: db, username: user, password: pass),
         result <- MyXQL.query!(pid, "SELECT * FROM some_table") do
      GenServer.stop(pid)

      result =
        result.rows
        |> Flow.from_enumerable()
        |> Flow.map(fn [_, field1, field2, _, _] -> {field1, field2} end)
        |> Flow.partition(key: &elem(&1, 0))
        |> Flow.reduce(fn -> %{} end, fn {field1, field2}, acc ->
          Map.update(
            acc,
            String.to_existing_atom(field1),
            [field2],
            &[field2 | &1]
          )
        end)

      Logger.info("Loaded #{Enum.count(result)} values from " <> host)

      {:ok, result}
    else
      :error ->
        Logger.warn("Skipped reconfig, one of MYSQL_{HOST,DB,USER,PASS} is missing")
        :ok

      error ->
        Logger.error("Reconfiguring failed. Error: " <> inspect(error))
        {:error, error}
    end
  end
end

  
    
  
  Callback Patterns


Kungfuig supports multiple callback mechanisms to notify your application when configuration changes:
Process Message Callback
# Using a PID - the process will receive a {:kungfuig_update, config} message
defmodule MyApp.ConfigConsumer do
  use GenServer
  
  def start_link(_), do: GenServer.start_link(__MODULE__, nil)
  
  def init(_), do: {:ok, %{}}

  # Handle configuration updates
  def handle_info({:kungfuig_update, config}, state) do
    IO.puts("Configuration updated: #{inspect(config)}")
    {:noreply, state}
  end
end

# In your application startup
{:ok, pid} = MyApp.ConfigConsumer.start_link([])
Kungfuig.start_link(workers: [{Kungfuig.Backends.Env, callback: pid}])
Callback Behaviour Implementation
defmodule MyApp.ConfigHandler do
  @behaviour Kungfuig.Callback
  
  @impl true
  def handle_config_update(config) do
    IO.puts("Configuration updated: #{inspect(config)}")
    :ok
  end
end

# In your application startup
Kungfuig.start_link(workers: [{Kungfuig.Backends.Env, callback: MyApp.ConfigHandler}])
GenServer Interaction Callbacks
# Using a GenServer cast
defmodule MyApp.ConfigManager do
  use GenServer

  def start_link(_), do: GenServer.start_link(__MODULE__, nil, name: __MODULE__)
  
  def init(_), do: {:ok, %{}}

  def handle_cast({:config_updated, config}, state) do
    # Process configuration update
    {:noreply, Map.put(state, :config, config)}
  end
end

# In your application startup
MyApp.ConfigManager.start_link([])
Kungfuig.start_link(
  workers: [
    {Kungfuig.Backends.Env, callback: {MyApp.ConfigManager, {:cast, :config_updated}}}
  ]
)
Function Callback
# Using an anonymous function
callback_fn = fn config ->
  IO.puts("Config updated: #{inspect(config)}")
  :ok
end

Kungfuig.start_link(workers: [{Kungfuig.Backends.Env, callback: callback_fn}])

  
    
  
  Validation System


Since v0.3.0, Kungfuig supports configuration validation through the Kungfuig.Validator behavior, which by default uses NimbleOptions for schema validation.
Creating a Custom Validator
defmodule MyApp.ConfigValidator do
  use Kungfuig.Validator, schema: [
    database: [
      type: :string,
      required: true,
      doc: "The database name"
    ],
    hostname: [
      type: :string,
      required: true,
      doc: "The database hostname"
    ],
    port: [
      type: :integer,
      default: 5432,
      doc: "The database port"
    ],
    ssl: [
      type: :boolean,
      default: false,
      doc: "Whether to use SSL"
    ]
  ]
end
Using the Validator
You can apply validators at two levels:
	Per backend - to validate only this backend's configuration:

Kungfuig.start_link(
  workers: [
    {MyApp.Kungfuig.DatabaseBackend, validator: MyApp.ConfigValidator}
  ]
)
	Global validator - to validate the entire configuration:

Kungfuig.start_link(
  validator: MyApp.GlobalConfigValidator,
  workers: [
    MyApp.Kungfuig.DatabaseBackend,
    MyApp.Kungfuig.RedisBackend
  ]
)

  
    
  
  Named Instances


Since v0.4.0, Kungfuig supports multiple named instances, allowing different parts of your application to use separate configuration managers.
# Start a Kungfuig instance for database configuration
Kungfuig.start_link(
  name: MyApp.DBConfig,
  workers: [
    {MyApp.Kungfuig.DatabaseBackend, interval: 60_000}
  ]
)

# Start another Kungfuig instance for API configuration
Kungfuig.start_link(
  name: MyApp.APIConfig,
  workers: [
    {MyApp.Kungfuig.APIBackend, interval: 30_000}
  ]
)

# Query the configurations separately
db_config = Kungfuig.config(:database, MyApp.DBConfig)
api_config = Kungfuig.config(:api, MyApp.APIConfig)

  
    
  
  Immediate Configuration Processing


Since v0.4.2, Kungfuig allows immediate configuration validation using the imminent: true option:
Kungfuig.start_link(
  workers: [
    {Kungfuig.Backends.Env, imminent: true}
  ]
)
This option ensures that the configuration is validated and processed during the init phase, rather than as a continuation after init returns. This is useful when your application requires the configuration to be available immediately upon startup.

  
    
  
  Testing


Simply implement a stub returning an expected config and you are all set.
defmodule MyApp.Kungfuig.Stub do
  @moduledoc false

  use Kungfuig.Backend

  @impl Kungfuig.Backend
  def get(_meta), do: %{foo: :bar, baz: [42]}
end

  
    
  
  Comparison with Alternatives


Kungfuig offers several advantages compared to other configuration management solutions:

  
    
  
  vs. Application Environment


The standard Elixir Application environment is static after startup unless manually changed:
	Kungfuig: Automatically polls for changes at configurable intervals
	App env: Requires manual calls to Application.put_env/3 to update


  
    
  
  vs. Config Providers


Elixir's built-in config providers run only once at application start:
	Kungfuig: Continually monitors for changes while the application is running
	Config providers: Only run during application startup


  
    
  
  vs. Distillery Config Providers


While Distillery's Config Providers can load configuration on application start:
	Kungfuig: Lets you define custom backends for any data source
	Distillery: Focused on files and environment variables
	Kungfuig: Supports multiple callback types for change notifications
	Distillery: Doesn't have built-in change notification


  
    
  
  vs. etcd/Consul/ZooKeeper clients


While these external services provide great configuration management:
	Kungfuig: Works with many different backends, not tied to a specific service
	External services: Require running and maintaining additional infrastructure
	Kungfuig: Simple supervision tree model familiar to Elixir developers
	External services: More complex client libraries and connection management


  
    
  
  Changelog


	1.0.0 — modern Elixir v1.16
	0.4.4 — fix a bug with hardcoded names (Supervisor and Blender)
	0.4.2 — allow imminent: true option to Kungfuig.Backend
	0.4.0 — allow named Kungfuig instances (thanks @vbroskas)
	0.3.0 — allow validation through NimbleOptions (per backend and global)
	0.2.0 — scaffold for backends + several callbacks (and the automatic one for Blender)

Kungfuig behaviour
    



      
Kungfuig provides a pluggable drop-in support for live configurations with change notification capabilities.

  
    
  
  Overview


Kungfuig (pronounced: [ˌkʌŋˈfig]) is a comprehensive configuration management system for Elixir applications
that provides:
	Live configuration updates from multiple sources
	Callback notifications when configurations change
	Configuration validation
	Extensible backend system for custom configuration sources
	Supervisor-managed configuration processes

This behaviour defines the dynamic config provider. In most cases, you don't need to implement
this behaviour directly. Instead, implement one or more backends (see Kungfuig.Backend)
and pass them, optionally parametrized, as workers: to start_link/1 as shown below:
{:ok, pid} =
  Kungfuig.start_link(
    workers: [
      {Kungfuig.Backends.EnvTransform,
       interval: 100,
       validator: Kungfuig.Validators.EnvTransform,
       callback: {self(), {:info, :updated}},
       callback: Kungfuig.Test.CallbackAlert}
    ]
  )
By default, Kungfuig starts two backends:
	Application environment backend (Application.get_all_env/1) 
	System environment backend (for environment variables)


  
    
  
  Architecture


Kungfuig operates using a supervisor tree pattern:
	A supervisor (Kungfuig.Supervisor) manages the configuration process
	A manager process (Kungfuig.Manager) supervises the backend workers
	Backend workers (Kungfuig.Backend implementations) poll their respective configuration sources
	A blender process (Kungfuig.Blender) combines configurations from all backends
	Callbacks are triggered when configurations change


  
    
  
  Custom Configuration Sources


If you need to get configuration updates from sources other than the default ones,
such as JSON/YAML files, Redis, databases, or any external service, you can create
your own backends by implementing the Kungfuig.Backend behaviour.
See Kungfuig.Backend for reference or check the 
Usage section in the README for an example
using MySQL.

  
    
  
  Validation


Since v0.3.0, Kungfuig supports validation through the Kungfuig.Validator behaviour,
which by default uses NimbleOptions for schema validation. Validators can be specified:
	Per backend (applying only to that backend's configuration)
	Globally (applying to the entire configuration)


  
    
  
  Callbacks


When a configuration changes, Kungfuig notifies interested parties through callbacks.
Multiple callback mechanisms are supported, as detailed in the callback/0 type.

  
    
  
  Named Instances


Since v0.4.0, Kungfuig supports multiple named instances, allowing different parts of your
application to use separate configuration managers.

  
    
  
  Examples


Basic usage:
# Start Kungfuig with default options
Kungfuig.start_link()

# Get all configuration
Kungfuig.config()
#⇒ %{env: %{kungfuig: []}, system: %{}}

# Get configuration from a specific backend
Kungfuig.config(:env)
#⇒ %{kungfuig: []}

# Update application environment and see the change
Application.put_env(:kungfuig, :foo, 42)
Kungfuig.config(:env)
#⇒ %{kungfuig: [foo: 42]}
Using callbacks:
# Using a process message as callback
Kungfuig.start_link(
  workers: [
    {Kungfuig.Backends.Env, callback: {self(), {:info, :config_updated}}}
  ]
)

# Using a module that implements Kungfuig.Callback
defmodule MyApp.ConfigHandler do
  @behaviour Kungfuig.Callback
  
  @impl true
  def handle_config_update(config) do
    IO.puts("Configuration updated: " <> inspect(config))
  end
end

Kungfuig.start_link(
  workers: [
    {Kungfuig.Backends.Env, callback: MyApp.ConfigHandler}
  ]
)
Using named instances (since v0.4.0):
# Start a named Kungfuig instance
Kungfuig.start_link(name: MyApp.Config)

# Query the named instance
Kungfuig.config(:env, MyApp.Config)
Using immediate validation (since v0.4.2):
# Start with imminent validation for immediate config processing
Kungfuig.start_link(
  workers: [
    {Kungfuig.Backends.Env, imminent: true, validator: MyApp.ConfigValidator}
  ]
)

        callback()

      


        The callback to be used for subscribing to configuration updates.



    


    
      
        config()

      


        The config map to be updated and exposed through callbacks.



    


    
      
        option()

      


        The options that can be passed to start_link/1 function.



    





  
    Functions
  


    
      
        config(which \\ :!, supervisor \\ Kungfuig)

      


        Retrieves the current config for the key(s) specified



    


    
      
        start_link()

      


        The function to start the Kungfuig manager under a supervision tree with default options



    


    
      
        start_link(opts)

      


        The function to start the Kungfuig manager under a supervision tree with custom options

      callback()



        
          
        

    

  


  

      

          @type callback() ::
  module()
  | pid()
  | {module(), atom()}
  | (config() -> :ok)
  | {GenServer.name() | pid(), {:call | :cast | :info, atom()}}


      


The callback to be used for subscribing to configuration updates.
Kungfuig supports several callback mechanisms:
	Module implementing Kungfuig.Callback behaviour:
defmodule MyApp.ConfigHandler do
  @behaviour Kungfuig.Callback
  
  @impl true
  def handle_config_update(config) do
    # Handle configuration update
    :ok
  end
end

	Process identifier (PID): The process will receive a {:kungfuig_update, config} message
# In a GenServer
def handle_info({:kungfuig_update, config}, state) do
  # Handle configuration update
  {:noreply, state}
end

	Module-function tuple: A tuple {module, function} where the function accepts a single 
argument (the configuration)
defmodule MyApp.ConfigHandler do
  def on_config_update(config) do
    # Handle configuration update
  end
end

# In your Kungfuig setup
callback: {MyApp.ConfigHandler, :on_config_update}

	Anonymous function: A function that takes one argument (the configuration)
callback: fn config -> 
  # Handle configuration update
  :ok 
end

	GenServer interaction tuple: A tuple specifying a GenServer and how to interact with it
# Send a cast to a GenServer
callback: {MyGenServer, {:cast, :config_updated}}

# Inside the GenServer
def handle_cast({:config_updated, config}, state) do
  # Handle configuration update
  {:noreply, state}
end



  



  
    
      
    
    
      config()



        
          
        

    

  


  

      

          @type config() :: %{required(atom()) => term()}


      


The config map to be updated and exposed through callbacks.
This is a map where the keys are atoms (typically representing backend names) and
the values are the configuration data from those backends.

  



  
    
      
    
    
      option()



        
          
        

    

  


  

      

          @type option() ::
  {:name, GenServer.name()}
  | {:callback, callback()}
  | {:interval, non_neg_integer()}
  | {:imminent, boolean()}
  | {:anonymous, boolean()}
  | {:start_options, GenServer.options()}
  | {:validator, module()}
  | {:workers, [module() | {module(), keyword()}]}


      


The options that can be passed to start_link/1 function.
Options include:
	:name - The name to register the Kungfuig instance under. Default is Kungfuig. 
@since v0.4.0

	:callback - One or more callbacks to notify when configuration changes. See callback/0.

	:interval - How frequently to check for configuration updates, in milliseconds. 
Default is 1000 (1 second).

	:imminent - When true, performs configuration validation immediately during initialization,
rather than as a continuation after init returns. Default is false.

	:anonymous - When true, the GenServer is not registered under a name.
Default is false.

	:start_options - Options passed to GenServer.start_link/3.

	:validator - A module implementing the Kungfuig.Validator behaviour to validate
the configuration. Default is Kungfuig.Validators.Void (no validation).
@since v0.3.0

	:workers - A list of backend specifications, each either a module name or a 
{module, opts} tuple. It might also be a function of arity 



  


        

      

      
        Functions

        


    

  
    
      
    
    
      config(which \\ :!, supervisor \\ Kungfuig)



        
          
        

    

  


  

      

          @spec config(which :: atom() | [atom()], supervisor :: Supervisor.supervisor()) ::
  config()


      


Retrieves the current config for the key(s) specified

  



  
    
      
    
    
      start_link()



        
          
        

    

  


  

The function to start the Kungfuig manager under a supervision tree with default options

  



  
    
      
    
    
      start_link(opts)



        
          
        

    

  


  

The function to start the Kungfuig manager under a supervision tree with custom options

  


        

      


  

    
Kungfuig.Backend behaviour
    



      
The scaffold for the backend watching the external config source.

  
    
  
  Example


defmodule Kungfuig.Backends.EnvTransform do
  @moduledoc false

  use Kungfuig.Backend, report: :logger, imminent: true

  @impl Kungfuig.Backend
  def get(meta, key) do
    {:ok,
     meta
     |> Keyword.get(:for, [key])
     |> Enum.reduce(%{}, &Map.put(&2, &1, Application.get_all_env(&1)))}
  end

  @impl Kungfuig.Backend
  def transform(%{kungfuig: env}) do
    {:ok, for({k, v} <- env, {_, value} <- v, do: {k, value})}
  end
end

      


      
        Summary


  
    Types
  


    
      
        key()

      


        The type for the key withing Kungfuig, typically an atom.



    





  
    Callbacks
  


    
      
        get(list)

      


        The implementation of the call to remote that retrieves the data



    


    
      
        get(list, key)

      


        The implementation of the call to remote that retrieves the data under a key specified



    


    
      
        key()

      


        The key this particular config would be stored under, defaults to module name



    


    
      
        report(any)

      


        The implementation of error reporting



    


    
      
        transform(any)

      


        The transformer that converts the retrieved data to internal representation

      key()



        
          
        

    

  


  

      

          @type key() :: term()


      


The type for the key withing Kungfuig, typically an atom.

  


        

      

      
        Callbacks

        


  
    
      
    
    
      get(list)



        
          
        

    

  


  

      

          @callback get([Kungfuig.option()]) :: {:ok, any()} | {:error, any()}


      


The implementation of the call to remote that retrieves the data

  



  
    
      
    
    
      get(list, key)



        
          
        

    

  


  

      

          @callback get([Kungfuig.option()], key()) :: {:ok, any()} | {:error, any()}


      


The implementation of the call to remote that retrieves the data under a key specified

  



  
    
      
    
    
      key()


        (optional)


        
          
        

    

  


  

      

          @callback key() :: key()


      


The key this particular config would be stored under, defaults to module name

  



  
    
      
    
    
      report(any)


        (optional)


        
          
        

    

  


  

      

          @callback report(any()) :: :ok


      


The implementation of error reporting

  



  
    
      
    
    
      transform(any)


        (optional)


        
          
        

    

  


  

      

          @callback transform(any()) :: {:ok, any()} | {:error, any()}


      


The transformer that converts the retrieved data to internal representation

  


        

      


  

    
Kungfuig.Callback behaviour
    



      
The behaviour defining the dynamic config provider’s callback.
Kungfuig accepts callbacks in several forms, see Kungfuig.callback/0.

      


      
        Summary


  
    Callbacks
  


    
      
        handle_config_update(config)

      


          The callback to be used for subscribing to config updates.



    





      


      
        Callbacks

        


  
    
      
    
    
      handle_config_update(config)



        
          
        

    

  


  

      

          @callback handle_config_update(Kungfuig.config()) :: :ok


      


  The callback to be used for subscribing to config updates.

  


        

      


  

    
Kungfuig.Validator behaviour
    



      
Generic validator encapsulation.
Default implementation is
  NimbleOptions.

  
    
  
  Example


defmodule TemperatureValidator do
  @moduledoc false

  use Kungfuig.Validator, schema: [
    city: [type: :string, required: true],
    temperature: [type: :integer, required: false]
  ]
end

    Callbacks
  


    
      
        doc(options)

      


        Generates a documentation for expected options



    


    
      
        validate(options)

      


        Validates given options with the schema, provided by the implementation of this behaviour



    





      


      
        Callbacks

        


  
    
      
    
    
      doc(options)



        
          
        

    

  


  

      

          @callback doc(options :: keyword()) :: String.t()


      


Generates a documentation for expected options

  



  
    
      
    
    
      validate(options)



        
          
        

    

  


  

      

          @callback validate(options :: map() | keyword()) ::
  {:ok, validated_options :: keyword()} | {:error, any()}


      


Validates given options with the schema, provided by the implementation of this behaviour

